```
Needs["Optica`Optica`"]
```

```
++++++++++++++++++++++++++++++++

Optica 3.0 was loaded in 7 s and needs
  11987 kilobytes of memory on top of 7246 kilobytes already used
```

Grating compressor defined in Backus et al. Rev. Sci. Instrum.,p. 1207. Vol 69, (1998).

```
CWL = 0.8; (*Center Wavelength [um]*)
GMM = 600;(* Grooves / mm *)
GS = 10;(* Grating Separation [mm] *)
TH = 13.89;(* incident angle [Degrees]*)
BW = 0.050; (*Bandwidth [um] *)
S1 = Move[SingleRay[WaveLength → CWL,
    SymbolicWaveLength → λ, IntrinsicMedium → Vacuum], {0, 0}, 180];
S2 = Move[RainbowOfRays[{CWL - BW / 2, CWL + BW / 2}, NumberOfRays → 11], {0, 0}, 180];
G1 = Move[GratingMirror[GMM, {1, 1}, DiffractedOrders → {{-1, 1.5}, {0, 0}, {1, 0}}],
   {0, 0}, (TH)];
G2 = Move[GratingMirror[GMM, {1, 1}, DiffractedOrders → {{-1, 0}, {0, 0}, {1, 1.5}}],
   {GS / Cos[TH °], 0}, (TH)];
R1 = Move[Mirror[{1, 1}], {-1, 0}];
B1 = Boundary[{-2, -1, -1}, {12, 1, 1}];
sys = {S1, G1, G2, R1, G2, G1, B1};
sys1 = {S2, G1, G2, R1, G2, G1, B1};
TurboPlot[sys1, PlotType → TopView]
```



```
-traced system-
```

Retrieve optical path length as a function of wavelength

```
res = TurboTrace[sys, OpticalLength, OutputType → RayTraceFunction, SequentialTrace → True]
```

```
{RayTraceFunction → RayTraceFunction[{λ}, -raytrace code: 1971224 Bytes- ],
 SymbolicValues → {λ → 0.8}, EmbedRays → True, EmbedThresholdIntensity → True,
 EmbedGenerationLimit → True, RayTracePrecision → 15.9546}
```

```
rayfunc = (RayTraceFunction /. res)
```

```
RayTraceFunction[{λ}, -raytrace code: 1971224 Bytes- ]
```

Calculate group delay as a function of wavelength and frequency

```
c = 299 792 458 000(*speed of light in mm/s*)
```

```
299 792 458 000
```

```
Delay[λ_] := rayfunc[λ] / c / 1*^-15 (* Group delay through the compressor in [fs] *)
Plot[Delay[λ] - Delay[CWL], {λ, CWL - BW / 2, CWL + BW / 2}, PlotRange → All,
 AxesLabel → {"Wavelength [μm]", "Relative Group Delay [fs]"}]
```



```
Frequ[λ_] := 2 π c / (λ * 1*^-3)
Wavel[ω_] := 2 π c / (ω * 1*^-3)

Plot[Delay[Wavel[ω]] - Delay[CWL], {ω, Frequ[CWL - BW / 2], Frequ[CWL + BW / 2]},
 PlotRange → All, AxesLabel → {"ω [Hz]", "Relative Group Delay [fs]"}]
```



Calculate Derivatives

```
f1 = Fit[Table[{ω, First[Delay[Wavel[ω]]]}, {ω, Frequ[CWL + BW / 2], Frequ[CWL - BW / 2],
      Frequ[CWL - BW / 2] - Frequ[CWL + BW / 2] / 100}], {1, x, x^2, x^3, x^4, x^5, x^6}, x];
Delayω[x_] := Evaluate[f1]
GVDω[x_] := Evaluate[∂x Delayω[x] / 1*^-15]
TODω[x_] := Evaluate[∂x,x Delayω[x] / 1*^-30]
FODω[x_] := Evaluate[∂x,x,x Delayω[x] / 1*^-45]
```

Compare Results to Table II p .1212 Rev. Sci. Instrum. Vol. 96 (1998)

The above calculations was for double - passing the compressor. For the comparison the results need to be divided by 2

```
TableForm[{{"GVD", GVDω[Frequ[0.8]] / 2 , "fs^2"},
  {"TOD", TODω[Frequ[0.8]] / 2 , "fs^3"}, {"FOD", FODω[Frequ[0.8]] / 2 , "fs^4"}}]
```

```
GVD  -3567.62  fs^2
TOD  5101.09   fs^3
FOD  -10225.7  fs^4
```

```
p1 = Plot[0.5 * (Delay[λ] - Delay[CWL]), {λ, CWL - BW / 2, CWL + BW / 2},
   PlotRange → All, AxesLabel → {"Wavelength [μm]", "Relative Group Delay [fs]"}];
p2 = Plot[0.5 * GVDω[Frequ[λ]], {λ, CWL - BW / 2, CWL + BW / 2},
   PlotRange → All, AxesLabel → {"Wavelength [μm]", "GVD [fs^2]"}];
p3 = Plot[0.5 * TODω[Frequ[λ]], {λ, CWL - BW / 2, CWL + BW / 2},
   PlotRange → All, AxesLabel → {"Wavelength [μm]", "TOD [fs^3]"}];
p4 = Plot[0.5 * FODω[Frequ[λ]], {λ, CWL - BW / 2, CWL + BW / 2},
   PlotRange → All, AxesLabel → {"Wavelength [μm]", "FOD [fs^4]"}];
Show[GraphicsArray[{{p1, p2}, {p3, p4}}, GraphicsSpacing → 0.05, Frame → True]]
```


Relative Group Delay [fs]
200
100
−100
−200
0.79
0.80
0.81
0.82
Wavelength [μm]
TOD [fs^3]
5500
5000
4500
0.79
0.80
0.81
0.82
Wavelength [μm]

Try Symbolic Calculation

```
symtrace = SymbolicTrace[sys, {λ > 0}, SeriesOrder → 6, MakeFloatingPoint → All,
   ReportedParameters → {SymbolicOpticalLength}, IncludeUserTerms → {λ},
   LocalCoordinateExpansions → None(*,RunningCommentary→True*)];
ol = SymbolicOpticalLength / c / 1*^-15 /. symtrace
Delaysym[λ_] := Evaluate[ol] (* Group delay in fs*)
```

$$\frac{1}{149\,896\,229}$$
$$500\,000\,000\,000\left(52.9322 - 0.839642\,\lambda + 6.84511\,\lambda^2 - 6.63417\,\lambda^3 + 7.20719\,\lambda^4 - 3.92741\,\lambda^5 + 1.12095\,\lambda^6\right)$$

```
Plot[Delaysym[λ] - Delaysym[CWL], {λ, CWL - BW / 2, CWL + BW / 2},
 PlotRange → All, AxesLabel → {"Wavelength [μm]", "Relative Group Delay [fs]"}]
```




```
Plot[Delaysym[Wavel[ω]] - Delaysym[CWL], {ω, Frequ[CWL - BW / 2], Frequ[CWL + BW / 2]},
 PlotRange → All, AxesLabel → {"ω [Hz]", "Relative Group Delay [fs]"}]
```



```
GVDωsym[x_] := Evaluate[ ∂x Delaysym[Frequ[x]] / 1*^-15 ]
TODωsym[x_] := Evaluate[ ∂x,x Delaysym[Frequ[x]] / 1*^-30 ]
FODωsym[x_] := Evaluate[ ∂x,x,x Delaysym[Frequ[x]] / 1*^-45 ]
```

Compare Results to Table II p .1212 Rev. Sci. Instrum. Vol. 96 (1998)

The above calculations was for double - passing the compressor. For the comparison the results need to be divided by 2

```
TableForm[{{"GVD", GVDωsym[Frequ[0.8]] / 2 , "fs^2"},
  {"TOD", TODωsym[Frequ[0.8]] / 2 , "fs^3"}, {"FOD", FODωsym[Frequ[0.8]] / 2 , "fs^4"}}]
```

```
GVD  -3567.62  fs^2
TOD  5101.09   fs^3
FOD  -10225.7  fs^4
```

```
p1s = Plot[0.5 * (Delaysym[λ] - Delaysym[CWL]), {λ, CWL - BW / 2, CWL + BW / 2},
   PlotRange → All, AxesLabel → {"Wavelength [μm]", "Relative Group Delay [fs]"}];
p2s = Plot[0.5 * GVDωsym[Frequ[λ]], {λ, CWL - BW / 2, CWL + BW / 2},
   PlotRange → All, AxesLabel → {"Wavelength [μm]", "GVD [fs^2]"}];
p3s = Plot[0.5 * TODωsym[Frequ[λ]], {λ, CWL - BW / 2, CWL + BW / 2},
   PlotRange → All, AxesLabel → {"Wavelength [μm]", "TOD [fs^3]"}];
p4s = Plot[0.5 * FODωsym[Frequ[λ]], {λ, CWL - BW / 2, CWL + BW / 2},
   PlotRange → All, AxesLabel → {"Wavelength [μm]", "FOD [fs^4]"}];
Show[GraphicsArray[{{p1s, p2s}, {p3s, p4s}}, GraphicsSpacing → 0.05, Frame → True]]
```


Relative Group Delay [fs]
200
100
−100
−200
0.79
0.80
0.81
0.82
Wavelength [μm]
TOD [fs^3]
5500
5000
4500
0.79
0.80
0.81
0.82
Wavelength [μm]

Compare numeric results to symbolic results

```
p1d = Plot[0.5 * ((Delaysym[λ] - Delaysym[CWL]) - (Delay[λ] - Delay[CWL])),
   {λ, CWL - BW / 2, CWL + BW / 2}, PlotRange → All,
   AxesLabel → {"Wavelength [μm]", "Delta Group Delay [fs]"}];
p2d = Plot[0.5 * (GVDωsym[Frequ[λ]] - GVDω[Frequ[λ]]), {λ, CWL - BW / 2, CWL + BW / 2},
   PlotRange → All, AxesLabel → {"Wavelength [μm]", "Delta GVD [fs^2]"}];
p3d = Plot[0.5 * (TODωsym[Frequ[λ]] - TODω[Frequ[λ]]), {λ, CWL - BW / 2, CWL + BW / 2},
   PlotRange → All, AxesLabel → {"Wavelength [μm]", "Delta TOD [fs^3]"}];
p4d = Plot[0.5 * (FODωsym[Frequ[λ]] - FODω[Frequ[λ]]), {λ, CWL - BW / 2, CWL + BW / 2},
   PlotRange → All, AxesLabel → {"Wavelength [μm]", "Delta FOD [fs^4]"}];
Show[GraphicsArray[{{p1d, p2d}, {p3d, p4d}}, GraphicsSpacing → 0.05, Frame → True]]
```


Delta Group Delay [fs]
1.×10^−6
5.×10^−7
−5.×10^−7
−1.×10^−6
0.79
0.80
0.81
0.82
Wavelength [μr
Delta TOD [fs^3]
0.010
0.005
−0.005
0.79
0.80
0.81
0.82
Wavelength [μr